Lecture 12

Introduction to Processing!

Processing

- Software toolkit that can be used for graphical programming, data visualization, physical computing, games, mobile, web and desktop applications, etc
- You can very quickly draw to the screen in code
- You can pretty much do *anything* with it

Processing

- Started around 2001
- based on Java (score!)
- https://processing.org/

Processing

- examples:
  - http://www.whiteglovetracking.com/
  - http://wearechopchop.com/%E2%80%9Cunnamed-soundsculpture%E2%80%9D/
  - http://www.niklasroy.com/project/88/my-little-piece-of-privacy/
  - http://superficie.ink/

Using Processing in Eclipse

- Download Processing
- follow this —> https://processing.org/tutorials/eclipse/
- We'll grab the Processing .jar file (which is kinda like a zip file with everything processing needs inside), and add it to our project

First Processing example

```
import processing.core.PApplet;

public class ProcessingTest extends PApplet {

    public static void main(String[] args) {
        PApplet.main("ProcessingTest");
    }

    public void settings(){
        size(300,300);
    }

    public void setup(){
        fill(120,50,240);
    }

    public void draw(){
        ellipse(mouseX, mouseY, 30, 30);
    }
}
```

First Processing example

- Things to note:
  - Class extends PApplet (We'll talk about this later)
  - All methods other than main are *not static*
  - main method has special line:

```
PApplet.main("ProcessingTest");
```

First Processing example

Every processing program we write will have these 3 methods:

```
/** set up size */
public void settings(){
   size(500, 500);
}
```

```
/** runs once at the beginning */
public void setup() {
   background(0);
}

/** runs over and over again as long as the program is still running */
public void draw() {
   ellipse(mouseX, mouseY, 30, 30);
}
```

First Processing example

Settings is used to set the width and height in pixels for our sketch

```
/** set up size */
public void settings(){
   size(500, 500);
}
```

First Processing example

Setup runs one time and prints to the screen.

```
/** runs once at the beginning */
public void setup() {
   background(0);
}
```

First Processing example

Once setup is done, draw runs *continuously* until the program is closed. Like a while loop. You don't *have* to use this, if you just want to draw once to the screen.

```
/** runs over and over again as long as the program is still running */
public void draw() {
   ellipse(mouseX, mouseY, 30, 30);
}
```

First Processing examples

- Check out the documentation: https://processing.org/reference/

- Look specifically at 2D Primitives, Color, and Input

First Processing examples - shapes

```
// x, y, width, height
ellipse(mouseX, mouseY, 30, 30);
```

```
// x, y, width, height
rect(mouseX, mouseY, 30, 30);

// x1, y1, x2, y2
line(mouseX, mouseY, 30, 30);
```

Practice: Draw a monster!

- Using simple shapes and different colors, draw a monster in Processing



First Processing examples - Interaction

- mousePressed and keyPressed booleans tell your program when the user clicks (you can also make a mousePressed() method)
- mouseX and mouseY are variables for the current cursor position.

```
if (mousePressed){
   ellipse(mouseX, mouseY, ellipseSize, ellipseSize);
}
```

```
if (keyPressed) {
   if (key == 'b') {
      fill(0);
   }
} else {
   fill(255);
}
ellipse(mouseX, mouseY, ellipseSize, ellipseSize);
```

First Processing examples -Instance variables

```
public class ProcessingTest extends PApplet {

   // Your methods can access these instance variables!
   int ellipseSize = 30;

...

   /** runs over and over again as long as the program is still running */
      public void draw() {
         if (keyPressed) {
            if (key == 'b') {
               fill(0);
            }
            if (keyCode == UP){
               ellipseSize++;
            }
         } else {
            fill(255);
         }
         ellipse(mouseX, mouseY, ellipseSize, ellipseSize);
      }
```

First Processing examples - animation

- How can we animate a shape?

- move a certain distance every frame

First Processing examples - animation

- How can we animate a shape?

- move a certain distance every frame

Working with pixels

- Each pixel in the Processing window is individually addressable!

- Processing has a pixels[] array that stores the color of each pixel

- you *must* load the pixels, manipulate them, then update the pixels in order to see changes

Working with pixels

```
public void setup(){
      loadPixels();
      for (int i = 0; i < pixels.length; i++){
         int randGrey = (int)(Math.random()*256);
         pixels[i] = color(randGrey);
      }

      updatePixels();
}
```

Working with pixels

- pixels[] is a one-dimensional array, so how can we get the “row” and “column” or coords of each pixel?

- index = row * width + columns

Working with pixels

```
public void setup(){
      loadPixels();
      for(int row = 0; row < height; row++) {
         for(int column = 0; column < width; column++) {
            int i = row * width + column;
            if (row % 10 == 0) {
               pixels[i] = color(255, 255, 255);
            } else {
               pixels[i] = color(0, 0, 0);
            }
         }
      }
      updatePixels();
}
```

For more information on Processing

- Check out the tutorials on Hello Processing: http://hello.processing.org/

- To dive deeper, check out
  - the nature of code: http://natureofcode.com/
  - the coding train: http://thecodingtrain.com/